Pomůcka pro cvičení: 2. semestr Bc studia

Numerické řešení integrálu-funkce daná tabelovanými hodnotami, funkce daná analyticky

1 LC - numerické řešení integrálu - rozšíření

Pro numerický výpočet určitého integrálu, kdy integrand je zadán analyticky, lze použít standardní příkaz **int** a následně (pokud Maple neumí najít přesný výsledek) příkaz **evalf**. Maple si sám zvolí vhodnou numerickou metodu (je použita hybridní symbolicko-numerická strategie). Je však možné zadat pomocí voleb příkazu **int** konkrétní metodu, přesnost atd. Detaily vizevalf,int.Jedná se o sofistikované profesionální metody, které nejsou obsahem základního kurzu. Tento postup je vhodný při řešení náročných reálných úloh, kdy jde zejména o rychlost a přesnost.

Pro pochopení numerických metod pro výpočet určitého integrálu, které se probírají v základním kurzu, je možné použít balíček, který nám umožní mimo jiné grafický výstup a animaci.

V tomto případě je potřeba načíst balíček **Student**[**Calculus1**] a v něm použít příkaz

Approximate Int(f(x), x=a..b, opts), kde **f(x)** je daná funkce, '<,>' (a, b) je interval, na kterém řešení hledáme, a **opts** jsou další parametry výstupu řešení (boxoptions, functionoptions, iterations, method, outline, output, partition, partitiontype, pointoptions, refinement, showarea, showfunction, showpoints, subpartition, title, view).

Příkazem **functionoptions** =*list*(list je seznam týkající se zobrazení funkce $f(x)$ je možn é m í nit zobrazen í funkce $f(x)$. Interně je nastaveno zobrazení plnou červenou čárou.

Příkazem **iterations** =*posint* (posint je kladné číslo) měníme počet zobrazení v animaci, interně je nastavena sekvence šesti obrázků.

Příkazem **method** =***lower*, *upper*, *left*, *midpoint*, *right*, *trapezoid*, *simpson, random*** nastavujeme aproximující metodu. Pokud nezadáme žádnou metodu, aproximace se provede metodou midpoint).

V Helpu je možné nalézt další numerické metody.

random

integrální součet s náhodně vybranými reprezentanty v každém intervalu

left

integrální součet s levými krajními body jako reprezentanty (levá obdélníková metoda)

midpoint

integrální součet, kde jako reprezentanty použijeme středy jednotlivých intervalů

right

integrální součet s pravými krajními body jako reprezentanty (pravá obdélníková metoda)

trapezoid

lichoběžníková metoda

simpson

Simpsonova metoda

Při použití **output** =***value*, *sum*,*plot*** nebo ***animation*** dostaneme výstup ve formě číselné hodnoty, sumy, grafu a animace. Pokud nezadáme žádný output, je výstupem číselná hodnota.

Nastavením **partition** =***posint*** (posint je kladné číslo) určíme počet podintervalů, na které interval '**<,>**' (a, b) rozdělíme. V Maple je interně nastaveno 10 podintervalů. POZOR! Dle je nutné použít příkaz **partitiontype**=***normal,*** jinak budeme u lichoběžníkové a Simpsonovy metody pracovat s dvojnásobným počtem intervalů, protože program má implicitně nastaveno, že podle dané metody jednotlivé podintervaly dělí na další 2 nebo 3.

Příkazem **pointoptions** =*list* (list je seznam týkající se zobrazení dělících bodů $\left(\frac{d}{dx}x\right)[i]$, $f\left(\left(\frac{d}{dx}x\right)[i]\right)$) je možné měnit zobrazení dělících bodů. Implicitně je nastaveno, že tyto body jsou zobrazeny zelenou barvou a mají tvar kroužku.

Příkazem **title** =*anything* zadáme libovolný název obrázku. Implicitní název závisí na metodě použité k nalezení aproximace. Další informace o specifikaci názvu vizplot/typesetting.

Příkaz **view** =[DEFAULT *or numeric..numeric*,DEFAULT *or numeric..numeric*] zobrazí (výřez) výsledného obrázku.

2 Fyzikální aplikace - numerické řešení

Př. Kyvadlo délky L má maximální výchylku θ_0. Použitím Newtonova druhého pohybového zákona bylo odvozeno, že perioda T kyvu kyvadla je dána vztahem

$$T = 4\sqrt{\frac{L}{g}}\int_0^{\frac{\pi}{2}} \frac{\mathrm{d}\,x}{\sqrt{1-k^2\sin^2 x}},$$

kde$k = 1/2$ (sin) θ_0 a g je tíhové zrychlení. Použitím Simpsonova pravidla (volte n=10) určete délku periody T, jestliže $L = m$ a $\theta_0 = 42°$.

```
> with(Student[Calculus1]):
```

Nejprve nadefinujeme proměnnéL, θ_0 a k, g budeme volit rovno 9,81$\overset{-2}{\mathrm{ms}}$.

```
> L:=1;
```

$$L := 1$$

```
> theta:=42*Pi/180;
```

$$\theta := \frac{7}{30}\pi$$

```
> k:=sin(theta/2);
```

$$k := \sin\left(\frac{7}{60}\pi\right)$$

```
> g:=9.81;
```

$$g := 9.81$$

Pokusíme se najít hodnotu T bez numerické integrace.

```
> T:=4*sqrt(L/g)*int(1/sqrt(1-k^2*sin(x)^2), x=0..Pi/2);
```

$$T := 1.277101714\,\mathit{EllipticK}\left(\sin\left(\frac{7}{60}\pi\right)\right)$$

Maple nám dává výsledek ve tvaru eliptického integrálu.
Použijeme tedy numerických metod k řešení dané úlohy. Nejprve si necháme vykreslit funkci $f(x)$ z integrandu. Pro lepší představu o jejím průběhu budeme volit na osách stejná měřítka (toho docílíme příkazem scaling=constrained), dále na ose x zvolíme dělící body jako násobky $1/8\,\pi$ (toho docílíme příkazem tickmarks = (spacing (Pi /8) , default) .

```
> f:=x->4*sqrt(L/g)*1/sqrt(1-k^2*sin(x)^2);
```

$$f := x \mapsto 4\sqrt{\frac{L}{g}}\frac{1}{\sqrt{1-k^2(\sin(x))^2}}$$

```
> plot(f(x),x=0..Pi/2,y=0..1.4,tickmarks = [spacing(Pi/8), default],
  scaling=constrained);
```



Příkazem ApproximateInt necháme spočítat hodnotu integrálu numericky.

```
> ApproximateInt(f(x),x=0..Pi/2);
```

$$2.075588702$$

```
> ApproximateInt(f(x),x=0..Pi/2,output=sum);
```

$$1/20\,\pi \sum_{i=0}^{9} 1.277101714 \left(\sqrt{1-\left(\sin\left(\frac{7}{60}\,\pi\right)\right)^2 (\sin(1/20\,(i+1/2)\,\pi))^2}\right)^{-1}$$

Dále si znázorníme pomocí ApproximateInt integrovanou oblast. Nenastavíme žádnou z numerických metod, takže Maple použije metodu midpoint. Výstup zobrazíme nejprve jako obrázek a pak jako animaci.

```
> ApproximateInt(f(x),x=0..Pi/2,tickmarks = [spacing(Pi/8), default],
  scaling=constrained,output=plot);
```




```
> ApproximateInt(f(x),x=0..Pi/2,tickmarks = [spacing(Pi/8), default],
  scaling=constrained,output=animation);
```



Nebudeme-li měnit počet dělících intervalů (je jich nastaveno deset), lze pro výpočet použít i Simpsonovy metody. Integrovanou oblast také znázorníme.

```
> ApproximateInt(f(x),x=0..Pi/2,method=simpson,partitiontype=normal);
```

$$2.075588704$$

```
> ApproximateInt(f(x),x=0..Pi/2,method=simpson,partitiontype=normal,
  tickmarks = [spacing(Pi/8), default],scaling=constrained,output=plot);
```



Pro srovnání lze použít i lichoběžníkové metody.

```
> ApproximateInt(f(x),x=0..Pi/2,method=trapezoid,partitiontype=normal);
```

$$2.075588703$$

```
> ApproximateInt(f(x),x=0..Pi/2,method=trapezoid,partitiontype=normal,
  tickmarks = [spacing(Pi/8), default],scaling=constrained,output=plot);
```



Můžeme použít také obdélníkovou metodu s náhodně generovanými zástupci v jednotlivých intervalech. Zde musíme vzít v úvahu, že po každém spuštění programu se nám náhodně vygenerují jiné body, takže hodnota integrálu bude pokaždé jiná (nebo musíme nastavit výchozí hodnotu generátoru).

```
> ApproximateInt(f(x),x=0..Pi/2,method=random);
```

$$2.075853850$$

```
> ApproximateInt(f(x),x=0..Pi/2,method=random,tickmarks = [spacing(Pi/8),
default],scaling=constrained, output=plot);
```



Pro srovnání uvedeme jednotlivé výsledky:

Obdélníková metoda: T=2.3144 s.

Simpsonova metoda: T=2.3143 s.

Lichoběžníková metoda: T= 2.3142 s.

Hodnoty jsou téměř stejné, což je dáno tím, že funkce je na uvažovaném intervalu téměř konstantní.

3 Numerické výpočty - funkce daná tabelovanými hodnotami

Kuldeep Sing: Engineering Mathematics through Applications, 433/4

Př. 1 Rychlost modelu nové lodi je testována v průběhu 5 sekund. Výsledky jsou uvedeny

v tabulce.

t(s)

0

1

2

3

4

5

$$v\left(ms^{-1}\right)$$

2.10

9.56

11.36

12.08

12.98

13.76

Dráha, kterou model urazí za 5 sekund, je dána vztahem $\int_0^5 vdt$. Vypočtěte hodnotu integrálu.

Pro grafické znázornění závislosti rychlosti na čase použijeme příkazu pointplot, který je součástí balíčku plot.

```
> restart;
> with(plots):
> with(Student[Calculus1]):
> data:=[[0,2.1],[1,9.56],[2,11.36],[3,12.08],[4,12.98],[5,13.76]];
```

$$data := [[0, 2.1], [1, 9.56], [2, 11.36], [3, 12.08], [4, 12.98], [5, 13.76]]$$

```
> pointplot(data,color=green,thickness = 5, symbol = solidcircle,
symbolsize = 30);
```



V našem případě nejde použít žádný ze zabudovaných příkazů balíčku Student[Calculus1], neboť Maple v tomto případě pracuje pouze se spojitými funkcemi. Zadání příkladu má ekvidistantní dělení

s délkou kroku h=1. Budeme muset použít některou z formulí: Složená obdélníková formule: $\int_a^b f(x)\, dx = h\,(y_0 + y_1 + ... + y_{n-1})$.

Složená lichoběžníková formule: $\int_a^b f(x)\, dx = \frac{h}{2}\,(y_0 + 2\, y_1 + ... + 2\, y_{n-1} + y_n)$.

Složená Simpsonova formule:
$\int_a^b f(x)\, dx = \frac{h}{3}\,(y_0 + y_{2\,m} + 4\,(y_1 + ... + y_{2\,m-1}) + 2\,(y_2 + ... + y_{2\,m-2}))$.

POZNÁMKA: Připomeňme, že u této metody musí být počet intervalů sudý, protože parabola, která funkci $f(x)$ nahrazuje, je určená trojicí bodů.

Složená obdélníková formule:

```
> Int(v,t=0..5)=2.1+9.56+11.36+12.08+12.98+13.76;
```

$$\int_0^5 v dt = 61.84$$

POZNÁMKA: Jiné způsoby odkazů na zadaná data. Jsou rychlejší, pokud zpracováváme data velkého rozsahu.

```
> Int(v,t=0..5)=data[1][2]+data[2][2]+data[3][2]+data[4][2]+data[5][2]
  +data[6][2];
```

$$\int_0^5 v dt = 61.84$$

```
> Int(v,t=0..5)=add(data[i][2],i=1..nops(data));
```

$$\int_0^5 v dt = 61.84$$

Složená lichoběžníková formule:

```
> Int(v,t=0..5)=1/2*(2.1+2*(9.56+11.36+12.08+12.98)+13.76);
```

$$\int_0^5 v dt = 53.91000000$$

POZNÁMKA: Jiné způsoby odkazů na zadaná data. Jsou rychlejší, pokud zpracováváme data velkého rozsahu.

```
> Int(v,t=0..5)=1/2*(data[1][2]+2*(data[2][2]+data[3][2]+data[4][2]
  +data[5][2])+data[6][2]);
```

$$\int_0^5 v dt = 53.91000000$$

```
> Int(v,t=0..5)=1/2*(data[1][2]+2*add(data[i][2],i=2..nops(data)-1)
  +data[nops(data)][2]);
```

$$\int_0^5 v dt = 53.91000000$$

POZNÁMKA: Složenou Simpsonovu formuli nelze použít, počet dělících intervalů není sudý.

Obdélníková metoda je obecně velmi jednoduchá, ale výsledky jsou velmi nepřesné.

Kuldeep Sing: Engineering Mathematics through Applications, 441/5 **Př. 2** Závislost proudu i na čase t je dána následující tabulkou

$t\,(s)$	$i\,(A)$
0	0,0
0,1	0,25
0,2	0,31
0,3	0,43
0,4	0,37
0,5	0,20
0,6	0,0

Střední hodnota proudui_{RMS} je dána vztahem$i_{RMS} = \sqrt{\frac{1}{0,6}\int_0^{0,6} i^2\left(t\right)\, dt}$. Nalezněte aproximaci proi_{RMS} .

Pro výpočet použijeme lichoběžníkovou metodu, protože obdélníková metoda nám dává velmi nepřesné výsledky.

Nejprve si znázorníme zadaná data.

```
> restart;
> with(plots):
> data1:=[[0,0],[0.1,0.25],[0.2,0.31],[0.3,0.43],[0.4,0.37],[0.5,0.2],[0.6,0]];
```

$$data1 := [[0,0],[0.1,0.25],[0.2,0.31],[0.3,0.43],[0.4,0.37],[0.5,0.2],[0.6,0]]$$

```
> pointplot(data1,color=green,symbol=solidcircle,symbolsize=20);
```



Pro úplnost znázorníme závislost $i^2(t)$.

```
> data2:=[[0,0],[0.1,0.25^2],[0.2,0.31^2],[0.3,0.43^3],[0.4,0.37^2],
> [0.5,0.2^2],[0.6,0]];
```

$$data2 := [[0,0],[0.1,0.0625],[0.2,0.0961],[0.3,0.079507],[0.4,0.1369],[0.5,0.04],[0.6,0]]$$

POZNÁMKA: Druhé mocniny druhých složek v zadaných datech lze získat i jiným zápisem.

```
> data3:=[seq([data1[i][1],data1[i][2]^2],i=1..nops(data1))];
```

$$data3 := [[0,0],[0.1,0.0625],[0.2,0.0961],[0.3,0.1849],[0.4,0.1369],[0.5,0.04],[0.6,0]]$$

```
> data4:=[seq([i[1],i[2]^2],i=data1)];
```

$$data4 := [[0,0],[0.1,0.0625],[0.2,0.0961],[0.3,0.1849],[0.4,0.1369],[0.5,0.04],[0.6,0]]$$

```
> pointplot(data2,color=green,symbol=solidcircle,symbolsize=20);
```



Stanovíme nejprve přibližnou hodnotu integrálu $\int_0^{0,6} i^2(t)\ dt$.

```
> Int(i^2,t=0..0.6)=0.1/2*(0+2*(0.25^2+.31^2+0.43^2+0.37^2+0.2^2)+0);
```

$$\int_0^{0.6} i^2 dt = 0.05204000000$$

```
> i_RMS=sqrt((1/0.6)*rhs(%));
```

$$i\_RMS = 0.2945052348$$

Pro srovnání použijeme i složenou Simpsonovu metodu, neboť počet dělících intervalů je sudý.

```
> Int(i^2,t=0..0.6)=0.1/3*(0+4*(0.31^2+0.37^2)+2*(0.25^2+0.43^2+0.2^2)+0);
```

$$\int_0^{0.6} i^2 dt = 0.05022666666$$

4 Numerické výpočty - funkce daná analyticky

Kuldeep Sing: Engineering Mathematics through Applications, 433/2b

Př. 1 Pomocí lichoběžníkové metody nalezněte aproximaci integrálu $\int_0^{\frac{\pi}{2}} \sqrt{\cos(x)}\ dx$. Interval rozdělte na čtyři podintervaly.

Nejprve zkusíme integrál vypočítat bez numerických metod.

```
> restart;
> Int(sqrt(cos(x)),x=0..Pi/2)=int(sqrt(cos(x)),x=0..Pi/2);
```

$$\int_0^{1/2\,\pi} \sqrt{\cos(x)} dx = -\sqrt{2} EllipticK\left(1/2\sqrt{2}\right) + 2\sqrt{2} EllipticE\left(1/2\sqrt{2}\right)$$

Výsledkem je eliptický integrál, příkazem evalf můžeme získat jeho číselnou hodnotu.

```
> evalf(int(sqrt(cos(x)),x=0..Pi/2));
```

$$1.198140235$$

Nyní provedeme numerickou integraci.

```
> with(Student[Calculus1]):
> Int(sqrt(cos(x)),x=0..Pi/2)=ApproximateInt(sqrt(cos(x)),x=0..Pi/2);
```

$$
\begin{aligned}
\int_0^{1/2\,\pi} \sqrt{\cos(x)}dx = {} & \frac{1}{80}\,\pi\,\sqrt{\sqrt{2}\sqrt{2-\sqrt{2}}+2\,\sqrt{2-\sqrt{2}}\sqrt{5+\sqrt{5}}-\sqrt{2}\sqrt{2-\sqrt{2}}\sqrt{5}-\sqrt{2}\sqrt{2+\sqrt{2}}+2\,\sqrt{2+\sqrt{2}}\sqrt{5+\sqrt{5}}+\sqrt{2}\sqrt{2+\sqrt{2}}\sqrt{5}} \\
& + \frac{1}{80}\,\pi\,\sqrt{2\,\sqrt{2}\sqrt{2-\sqrt{2}}\sqrt{5-\sqrt{5}}+2\,\sqrt{2+\sqrt{2}}\sqrt{5}+2\,\sqrt{2+\sqrt{2}}} \\
& + 1/40\,\pi\,\sqrt{2}\sqrt[4]{2+\sqrt{2}}+\frac{1}{80}\,\pi\,\sqrt{2\,\sqrt{2-\sqrt{2}}\sqrt{5}+2\,\sqrt{2-\sqrt{2}}+2\,\sqrt{2}\sqrt{2+\sqrt{2}}\sqrt{5-\sqrt{5}}} \\
& + \frac{1}{80}\,\pi\,\sqrt{-\sqrt{2}\sqrt{2-\sqrt{2}}+2\,\sqrt{2-\sqrt{2}}\sqrt{5+\sqrt{5}}+\sqrt{2}\sqrt{2-\sqrt{2}}\sqrt{5}+\sqrt{2}\sqrt{2+\sqrt{2}}+2\,\sqrt{2+\sqrt{2}}\sqrt{5+\sqrt{5}}-\sqrt{2}\sqrt{2+\sqrt{2}}\sqrt{5}} \\
& + \frac{1}{80}\,\pi\,\sqrt{2\,\sqrt{2}\sqrt{2-\sqrt{2}}\sqrt{5+\sqrt{5}}+2\,\sqrt{2+\sqrt{2}}\sqrt{5}-2\,\sqrt{2+\sqrt{2}}} \\
& + \frac{1}{80}\,\pi\,\sqrt{2\,\sqrt{2-\sqrt{2}}\sqrt{5-\sqrt{5}}+\sqrt{2}\sqrt{2-\sqrt{2}}\sqrt{5}+\sqrt{2}\sqrt{2-\sqrt{2}}-2\,\sqrt{2+\sqrt{2}}\sqrt{5-\sqrt{5}}+\sqrt{2}\sqrt{2+\sqrt{2}}\sqrt{5}+\sqrt{2}\sqrt{2+\sqrt{2}}} \\
& + 1/40\,\pi\,\sqrt{2}\sqrt[4]{2-\sqrt{2}} \\
& + \frac{1}{80}\,\pi\,\sqrt{2\,\sqrt{2-\sqrt{2}}\sqrt{5-\sqrt{5}}+\sqrt{2}\sqrt{2-\sqrt{2}}\sqrt{5}+\sqrt{2}\sqrt{2-\sqrt{2}}+2\,\sqrt{2+\sqrt{2}}\sqrt{5-\sqrt{5}}-\sqrt{2}\sqrt{2+\sqrt{2}}\sqrt{5}-\sqrt{2}\sqrt{2+\sqrt{2}}} \\
& + \frac{1}{80}\,\pi\,\sqrt{2\,\sqrt{2}\sqrt{2-\sqrt{2}}\sqrt{5+\sqrt{5}}-2\,\sqrt{2+\sqrt{2}}\sqrt{5}+2\,\sqrt{2+\sqrt{2}}}
\end{aligned}
$$

```
> rhs(evalf(%));
```

$$1.201931792$$

Lichoběžníková metoda.

```
> Int(sqrt(cos(x)),x=0..Pi/2)=ApproximateInt(sqrt(cos(x)),x=0..Pi/2,
  method=trapezoid,partition=4,partitiontype=normal);
```

$$\int_0^{1/2\,\pi}\sqrt{\cos(x)}dx = 1/16\,\pi\,\left(1+\sqrt{2}\sqrt[4]{2+\sqrt{2}}+2^{3/4}+\sqrt{2}\sqrt[4]{2-\sqrt{2}}\right)$$

```
> rhs(evalf(??));
```

$$1.146955054$$

Díky sudému počtu podintervalů možno použít Simpsonovy metody.

```
> Int(sqrt(cos(x)),x=0..Pi/2)=ApproximateInt(sqrt(cos(x)),x=0..Pi/2,
  method=simpson,partition=4,partitiontype=normal);
```

$$\int_0^{1/2\,\pi}\sqrt{\cos(x)}dx = 1/24\,\pi\,\left(1+2\sqrt{2}\sqrt[4]{2+\sqrt{2}}+2^{3/4}+2\sqrt{2}\sqrt[4]{2-\sqrt{2}}\right)$$

```
> rhs(evalf(??));
```

$$1.178227545$$

Kuldeep Sing: Engineering Mathematics through Applications, 442/8

Př. 2 Síla F působící na nosník o délce 5 m je dána vztahem$F = \displaystyle\int_0^5 \frac{e^x}{1+\sqrt{x}}dx$, kde x je vzdálenost. Vypočtěte velikost síly F.

Nejprve si znázorníme daný integrand.

```
> f:=x->exp(x)/(1+sqrt(x));
```

$$f := x \mapsto \frac{e^x}{1+\sqrt{x}}$$

Zobrazíme integrand.

```
> plot(f(x), x=0..5);
```



Při pokusu o výpočet bez numerické integrace dochází k opisování zadaného integrálu, což značí, že Maple neumí integrál vyřešit bez použití numerických metod.

```
> Int(f(x),x=0..5)=int(f(x),x=0..5);
```

$$\int_0^5 \frac{e^x}{1+\sqrt{x}}dx = \int_0^5 \frac{e^x}{1+\sqrt{x}}dx$$

```
> Int(f(x),x=0..5)=ApproximateInt(f(x),x=0..5,method=simpson,partition=8,
  partitiontype=normal);
```

$$\frac{\cdots e^{5/2}\sqrt{3}\sqrt{7} + 37280\, e^{5/8}\sqrt{3}\sqrt{7} + 7560\,\sqrt{5}\sqrt{3}\sqrt{7} + 4920\, e^{5/4}\sqrt{5}\sqrt{3}\sqrt{2}\sqrt{7} + 12720\, e^{\frac{15}{8}}\sqrt{5}\sqrt{3}\sqrt{2}\sqrt{7} + 7680\, e^{\frac{25}{8}}\sqrt{2}\sqrt{5}\sqrt{3}\sqrt{7} + 4380\, e^{\frac{15}{4}}\sqrt{2}\sqrt{5}\sqrt{3}\cdots}{\left(2+\sqrt{2}\sqrt{5}\right)\left(4+5\sqrt{2}\right)\left(2+\sqrt{5}\sqrt{3}\right)}$$

```
> rhs(evalf(??));
```

$$49.87773147$$

Zobrazíme plochu ohraničenou funkcí *f*.

```
> ApproximateInt(f(x),x=0..5,method=simpson,partition=8,
  partitiontype=normal,output=plot);
```




```
> Int(f(x),x=0..5)=ApproximateInt(f(x),x=0..5,method=simpson,
  partition=2,partitiontype=normal);
```

$$\int_0^5 \frac{e^x}{1+\sqrt{x}}dx = 5/6\, \frac{2\sqrt{5}+2+5\sqrt{2}+\sqrt{2}\sqrt{5}+8\,e^{5/2}\sqrt{5}+8\,e^{5/2}+2\,e^5+e^5\sqrt{2}\sqrt{5}}{\left(2+\sqrt{2}\sqrt{5}\right)\left(\sqrt{5}+1\right)}$$

```
> ApproximateInt(f(x),x=0..5,method=simpson,partition=2,
  partitiontype=normal,output=plot);
```



Výpočet pomocí Simpsonovy metody.

```
> Int(f(x),x=0..5)=(5/6)*(f(0)+f(5)+4*f(5/2));
```

$$\int_0^5 \frac{e^x}{1+\sqrt{x}}dx = 5/6 + 5/6\,\frac{e^5}{\sqrt{5}+1} + 10/3\,\frac{e^{5/2}}{1+1/2\,\sqrt{10}}$$

Srovnání výsledků získaných pomocí zabudované Simpsonovy metody a ruční editací.

```
> rhs(evalf[20](??)); rhs(evalf[20](??));
```

$$54.784535377042959746$$

$$54.784535377042959744$$